How to Solve the Transportation Problem with MPL/CPLEX
By your software advisor Gabriel Plano

Problem 1
Suppose you have three canneries (sources) and four warehouses (destinations). The shipment costs, outputs (1 for each source) and demands (1 for each destination) are as follows:

		Warehouses				
		1	2	3	4	Output
Canneries	1	464	513	654	867	75
	2	352	416	690	791	125
	3	995	682	388	685	100
	Demand	80	65	70	85	

Open MPL, go to file -> new then type the following:

```
index
        source := 1..3;                                      ! equivalent to writing "source := (1,2,3);"
        destination := 1..4;
data
        Shipcost[source,destination]:= (464,513,654,867,
                                         352,416,690,791,
                                         995,682,388,685);
        Output[source]:= (75,125,100);                       ! the output for the 3 sources
        Demand[destination]:= (80,65,70,85);                 ! the demand for the 4 destinations
variables
        Ship[source,destination];                            ! these are the 3*4 = 12 decision variables
model
        min C = sum(source,destination: Shipcost*Ship);      ! the sum is over both indices
subject to
        C1[source]:                                  ! constraint 1
                sum(destination: Ship) = Output;     ! the total shipped by each cannery equals its output
        C2[destination]:                             ! constraint 2
                sum(source: Ship) = Demand;          ! the total shipped to each warehouse equals its demand
end
```

The summation in constraint 1 is like summing over the columns in the table above. You take the sum over the destination index, so this is the index that you specify in the sum command: "sum(**destination**: Ship)". Similarly, for constraint 2 the sum is over the source index, like summing over rows in the table above.

To solve the problem, go to Run -> Solve -> Solve CPLEX 300, or press .

You should get the following optimal solution:

```
MIN C       =   152535.000000

      source    destination     Activity
  -------------------------------------
        1           1         0.000000
        1           2        20.000000
        1           3         0.000000
        1           4        55.000000
        2           1        80.000000
        2           2        45.000000
        2           3         0.000000
        2           4         0.000000
        3           1         0.000000
        3           2         0.000000
        3           3        70.000000
        3           4        30.000000
  -------------------------------------
```

Problem 2
Now, here is how you would solve a problem with a dummy source and big M's in MPL. We use the water allocation example from Chapter 8. Specifically, you should look at table 8.12 on page 318. Type the following in MPL.

```
index
        source := (Colombo,Sacron,Calorie,Dummy);                            ! sources are rivers
        destination := (B_min,B_extra,Los_Devils,San_Go,Hollyglass);         ! destinations are cities
data
        DeliveryCost[source,destination] := ( 16,16,13,22,17,
                                              14,14,13,19,15,
                                              19,19,20,23,-1,         ! replace the M's from table 8.12
                                              -1,0 ,-1, 0, 0);        ! with -1's
        Supply[source] := (50 60 50 50);                              ! from the column labeled supply
        Demand[destination] := (30 20 70 30 60);                      ! from the row labeled demand
variables
```

```
        Ship[source,destination]
                where (DeliveryCost[source,destination] >= 0);      ! this is the only new line
model
        min cost = sum(source,destination: DeliveryCost*Ship);
subject to
        C1[source]:
                sum(destination: Ship) = Supply;
        C2[destination]:
                sum(source: Ship) = Demand;
end
```

The only difference is that M’s are replaced with -1’s and that the line “where (DeliveryCost[source,destination] >= 0);” must be added when defining the Ship[source,destination] variable. This line makes it so that the impossible “routes” from rivers to cities are not among the decision variables, as these routes do not satisfy DeliveryCost[source,destination] >= 0.
Run CPLEX and you should get the following solution:

```
MIN cost     =     2460.000000

   source   destination          Activity
  --------------------------------------
   Colombo  B_min                0.000000
   Colombo  B_extra              0.000000
   Colombo  Los_Devils          50.000000
   Colombo  San_Go               0.000000
   Colombo  Hollyglass           0.000000
   Sacron   B_min                0.000000
   Sacron   B_extra              0.000000
   Sacron   Los_Devils          20.000000
   Sacron   San_Go               0.000000
   Sacron   Hollyglass          40.000000
   Calorie  B_min               30.000000
   Calorie  B_extra             20.000000
   Calorie  Los_Devils           0.000000
   Calorie  San_Go               0.000000
   Dummy    B_extra              0.000000
   Dummy    San_Go              30.000000
   Dummy    Hollyglass          20.000000
  --------------------------------------
```

This is consistent with the solution given in table 8.23 on page 333.